NAG Library Routine Document

F01ENF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F01ENF computes the principal matrix square root, $A^{1/2}$, of a real n by n matrix A.

2 Specification

```
SUBROUTINE F01ENF (N, A, LDA, IFAIL)

INTEGER            N, LDA, IFAIL
REAL (KIND=nag_wp) A(LDA,*)
```

3 Description

A square root of a matrix A is a solution X to the equation $X^2 = A$. A nonsingular matrix has multiple square roots. For a matrix with no eigenvalues on the closed negative real line, the principal square root, denoted by $A^{1/2}$, is the unique square root whose eigenvalues lie in the open right half-plane.

$A^{1/2}$ is computed using the algorithm described in Higham (1987). This is a real arithmetic version of the algorithm of Björck and Hammarling (1983). In addition a blocking scheme described in Deadman *et al.* (2013) is used.

4 References

Björck Å and Hammarling S (1983) A Schur method for the square root of a matrix *Linear Algebra Appl.* **52/53** 127–140

Deadman E, Higham N J and Ralha R (2013) Blocked Schur Algorithms for Computing the Matrix Square Root *Applied Parallel and Scientific Computing: 11th International Conference, (PARA 2012, Helsinki, Finland)* P. Manninen and P. Öster, Eds *Lecture Notes in Computer Science* **7782** 171–181 Springer–Verlag

Higham N J (1987) Computing real square roots of a real matrix *Linear Algebra Appl.* **88/89** 405–430

Higham N J (2008) *Functions of Matrices: Theory and Computation* SIAM, Philadelphia, PA, USA

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $N \geq 0$.

2: $A(LDA,*)$ – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least N.

On entry: the n by n matrix A.

On exit: contains, if $IFAIL = 0$, the n by n principal matrix square root, $A^{1/2}$. Alternatively, if $IFAIL = 1$, contains an n by n non-principal square root of A.

3: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F01ENF is called.

Constraint: LDA $\geq$ N.

4: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

A has a semisimple vanishing eigenvalue. A non-principal square root is returned.

IFAIL $= 2$

A has a defective vanishing eigenvalue. The square root cannot be found in this case.

IFAIL $= 3$

A has a negative real eigenvalue. The principal square root is not defined. F01FNF can be used to return a complex, non-principal square root.

IFAIL $= 4$

An internal error occurred. It is likely that the routine was called incorrectly.

IFAIL $= -1$

On entry, N $= \langle value \rangle$.
Constraint: N ≥ 0.

IFAIL $= -3$

On entry, LDA $= \langle value \rangle$ and N $= \langle value \rangle$.
Constraint: LDA $\geq$ N.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

The computed square root $\hat{X}$ satisfies $\hat{X}^2 = A + \Delta A$, where $\|\Delta A\|_F \approx O(\epsilon)n^3\|\hat{X}\|_F^2$, where ϵ is ***machine precision***.

For further discussion of the condition of the matrix square root see Section 6.1 of Higham (2008).

8 Parallelism and Performance

F01ENF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F01ENF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The cost of the algorithm is $85n^3/3$ floating-point operations; see Algorithm 6.7 of Higham (2008). $O(2 \times n^2)$ of real allocatable memory is required by the routine.

If condition number and residual bound estimates are required, then F01JDF should be used.

10 Example

This example finds the principal matrix square root of the matrix

$$A = \begin{pmatrix} 507 & 622 & 300 & -202 \\ 237 & 352 & 126 & -60 \\ 751 & 950 & 440 & -286 \\ -286 & -326 & -192 & 150 \end{pmatrix}.$$

10.1 Program Text

```
    Program f01enfe

!     F01ENF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: f01enf, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, lda, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:)
!     .. Executable Statements ..
      Write (nout,*) 'F01ENF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
```

```
      Read (nin,*)
      Read (nin,*) n

      lda = n
      Allocate (a(lda,n))

!     Read A from data file
      Read (nin,*)(a(i,1:n),i=1,n)

!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0

!     Find sqrt(A)
      Call f01enf(n,a,lda,ifail)

!     Print solution
      If (ifail==0) Then
        ifail = 0
        Call x04caf('G','N',n,n,a,lda,'sqrt(A)',ifail)
      End If

    End Program f01enfe
```

10.2 Program Data

```
F01ENF Example Program Data

4                                     :Value of N

 507.0     622.0     300.0     -202.0
 237.0     352.0     126.0      -60.0
 751.0     950.0     440.0     -286.0
-286.0    -326.0    -192.0      150.0  :End of matrix A
```

10.3 Program Results

```
 F01ENF Example Program Results

 sqrt(A)
              1            2            3            4
 1    1.5000E+01   1.4000E+01   8.0000E+00  -6.0000E+00
 2    6.0000E+00   1.4000E+01   3.0000E+00   9.6589E-15
 3    2.1000E+01   2.4000E+01   1.2000E+01  -8.0000E+00
 4   -5.0000E+00  -4.0000E+00  -7.0000E+00   8.0000E+00
```